extends Node2D

## 假设这是一个作物或者其他什么放置收益的单位

## 当前收益
var income : int = 0:
	set(value):
		income = value
		$VBoxContainer/Label.text = "当前收益：" + str(income)
## 收益频率
@export var frequency: float = 5
## 单次收益数量
@export var income_count: int = 1
## 最后一次记录的时间戳
var _time: float = 0
## 当前记录的时间差
var _difference: float = 0

func _ready() -> void:
	$VBoxContainer/Label2.text = "假设作物每 {frequency} 秒收货 {income_count}".format({"income_count": income_count, "frequency": frequency})

func _process(_delta: float) -> void:
	_difference += TimeManager.time - _time			# 记录时间差，在原有时间差基础上增加
	var quotient : int = _difference / frequency  	# 整除，得到商的整数部分（也就是根据时间差计算收益次数）
	_difference = fmod(_difference, frequency)  	# 取余，得到余数（如果单次时间差不够收益或者有剩余需要记录）
	income += quotient * income_count				
	_time = TimeManager.time						# 记录当前时间戳
